


# The Design of an Experimental Programming Language and its Translator

The Nuua Programming Language

*ÈRIK CAMPOBADAL FORÉS*

A bachelor thesis submitted to fulfill the degree of  
ICT Systems Engineer



Department of Mining, Industrial and ICT Engineering  
Advisor: Sebastia Vila Marta  
Polytechnic University of Catalonia  
Catalonia, Spain  
June 2019

# Contents

<b>1</b>	<b>Introduction</b>	<b>4</b>
1.1	Objectives . . . . .	4
1.2	Preliminary overview . . . . .	5
1.2.1	Language grammar . . . . .	5
1.2.2	Compilers . . . . .	6
1.2.3	Phases of a compiler . . . . .	7
1.3	Interpreters . . . . .	9
1.4	System architecture . . . . .	10
<b>2</b>	<b>The Nuua Language</b>	<b>13</b>
2.1	Grammar . . . . .	13
2.1.1	Lexical Grammar . . . . .	13
2.1.2	Syntax Grammar . . . . .	15
2.1.3	Operator Precedence . . . . .	22
2.2	Data types . . . . .	23
2.2.1	Integers . . . . .	23
2.2.2	Floats . . . . .	24
2.2.3	Booleans . . . . .	24
2.2.4	Strings . . . . .	24
2.2.5	Lists . . . . .	24
2.2.6	Dictionaries . . . . .	25
2.2.7	Functions . . . . .	25
2.2.8	Objects . . . . .	26
2.3	Statements . . . . .	26
2.3.1	Use Declaration . . . . .	26
2.3.2	Function Declaration . . . . .	27
2.3.3	Class Declaration . . . . .	28
2.3.4	Export Declaration . . . . .	29
2.3.5	Variable Declaration . . . . .	29
2.4	Expressions . . . . .	30
<b>3</b>	<b>Forthcoming Features</b>	<b>31</b>
<b>4</b>	<b>Bibliography</b>	<b>32</b>

# List of Figures

1.1	Compiler overview . . . . .	7
1.2	Common compiler phases . . . . .	8
1.3	Layered system . . . . .	10
1.4	Nuua's architecture (Layered System) . . . . .	12

# 1 Introduction

Programming languages are used every day by millions of engineers as part of their daily routine. A programming language is used to tell a computer what to do. When somebody wants a computer to do something, it needs to write a program using a programming language. Then a compiler needs to translate it into machine code for the computer to execute it.

To design a programming language it's important to understand the theory behind a compiler and learn about all the different steps involved to make a computer understand and execute a programming language.

## 1.1 Objectives

The main objective of this thesis is to design an experimental programming language and implement an interpreter to execute any program written with it. The different challenges that are faced during the design and implementation process are also explained and solved in their respective chapters. The experimental language built in this thesis is called **Nuua**.

The objective can be partitioned into the following points.

- Learn all the steps involved in a common compiler implementation and reproduce them according to the project needs.
- Design the Nuua Programming Language. This grammar must be simple, elegant and yet it needs to follow the most common programming language's specifications to avoid confusion when learning it, allowing either mature programmers or newcomers to pick up the grammar quickly and efficiently.
- Choose an efficient programming language to build the compiler with. Among other options, the languages that satisfy the previous statement are low-level programming languages like C, C++, D, Rust, Go, Crystal, Zig or Nim.
- Define a scalable software system to allow the compiler to grow efficiently. The software system that is chosen will be very important towards the scalability of the project.
- Build a compiler for the Nuua programming language and a very simple standard library.

## 1.2 Preliminary overview

This section details the most important aspects found in a compiler. This preliminary overview won't deal with details and only explains the basics to understand the whole system without deep knowledge. Further chapters contain expanded information respective to some of the details mentioned here.

### 1.2.1 Language grammar

Context-free grammar is a notation used to specify the syntax of a programming language. Following the syntax definition explained in [2, Section 2.2] a context-free grammar consists of four components:

1. A group of terminal symbols also known as tokens. In a programming language tokens may be literal symbols like '+', '\*' or numbers and identifiers.
2. A group of non-terminals that can be reduced to terminals based on the production rules.
3. A group of production rules that consists of a non-terminal on the left side and a sequence of terminals and/or non-terminals on the right side.
4. A non-terminal start symbol.

This thesis will use the *extended Backus-Naur form* also known as *EBNF* to express the context-free grammar representation of Nuua. EBNF is often used in different ways due to the big ammount of variants that exists. To clarify the EBNF syntax that this thesis is going to use, the following table will be used to specify the production rules. More information may be found in the EBNF grammar article [7].

Symbol	Definition
:	Used to define a production rule.
<i>space</i>	Used to concatenate patterns (space separated).
A B	Used to define a union of A and B.
A+	Used to define a one or more pattern of A.
A*	Used to define a zero or more pattern.
A?	Used to define an optional pattern.
(A)	Used to group a pattern.
"T" or 'T'	Used to define a terminal symbol.
@A	Used to indicate anything except A.
;	Used to terminate a given production rule.

Table 1.1: Variation of EBNF syntax used by this thesis

As a simple example, to define a language that consists in a single integer, the following EBNF grammar could be used:

```
integer
:  ("-" | "+")? digit+
;
digit
:  "0"|"1"|"2"|"3"|"4"|"5"|"6"|"7"|"8"|"9"
;
```

This variation is often used by many parser generators since it introduces a more visible and versatile approach to write the language grammar.

## 1.2.2 Compilers

The job of a compiler is to take an input language and translate it into another as shown in Figure 1.1. Often, the compiler term is used to express a translation to a much different level of abstraction, that usually means that the input is written in a high-level language and further translated into another low-level language. However, there are multiple types of compilers, and each one is used to implement different translation tasks. Nuua's system architecture built in this thesis, as described in section 1.4, is based on an interpreter architecture, which consists of a compiler that translates the Nuua language into an intermediate representation known as *bytecode* and further interpreted by a virtual machine.



Figure 1.1: Compiler overview

### 1.2.3 Phases of a compiler

A compiler can also be decoupled into different parts. Each part does a very different job but they are all connected to each other. In a typical compiler architecture, we may find all the different phases described in Figure 1.2.

Those phases are often found to be different depending on the implementation of the language. However, it's important to note what they do, since they are often implemented in one way or another. More implementation details are explained in their respective chapters but in this section, a small introduction to each phase is needed to understand the Nuua's system.

- **Lexical analysis:** In this phase, the input source is transformed from a character string into a token list, this is also called tokenization. Tokens are known and have certain attributes. For example, some tokens might include integers, symbols (+, -, \*, etc.) or identifiers among others. Lexemes are also evaluated as individual tokens, making a single token for identifiers matched as language keywords (like 'if', 'while', etc.). This phase is also called a lexer or scanner.
- **Syntax analysis:** In this phase, the implementation may vary among compilers, some of them work close to the lexical analysis since they can work together. However, its purpose is to perform operations to the token list to parse the input and create an Abstract Syntax Tree (AST<sup>1</sup>). An AST is already a structural representation that represents your input source. This stage determines if it is a valid program based on the language grammar and the specified rules. There are also scanner-less parsers that takes the lexical analysis and the syntax analysis into a single step. It is harder to understand and build but it often leads to faster parsing and less memory usage.
- **Semantic analysis:** This phase analyzes the AST and creates a symbol table while analyzes the input source with things like type checking or variable declarations, if some operations can be performed (for example adding a number with a string). A symbol table is a structure used by further phases to see information attached to specific source code parts. For example, it can store information about a variable (if it's global, exported, etc.).
- **Intermediate code:** An intermediate representation (IR) can be avoided but it's

---

<sup>1</sup>Abstract Syntax Tree (Wikipedia)




Figure 1.2: Common compiler phases



often used to have a platform independent optimizer. Usually the code generation targets a specific architecture and would require different optimizers depending on each architecture. However, by having an IR it's possible to have a single optimizer. A much used IR is Three Address Code (TAC<sup>2</sup>) that can be organized in quadruples or triples.

- **Optimization:** This optimization is often performed on the IR and performs different tasks to allow a faster and smaller output. For example, it may remove dead code, perform loop optimizations, etc.
- **Code generation:** This is where the real translation takes place, it translates the IR into a different language output. For example machine code. This phase often has to deal with instruction scheduling or register allocation while they have to output a fully working program.

## 1.3 Interpreters

There are different ways to interpret a language. This thesis implements a bytecode virtual machine (register-based) where the bytecode is the intermediate representation generated by the Nuua compiler. Among the most common interpreters built for programming languages, we may find the following types of interpreters:

- **Bytecode interpreter:** The programming language is first compiled into an intermediate bytecode representation and further executed by this virtual machine. A very famous example of a programming language interpreted by virtual machine is the Lua virtual machine [14]
- **Abstract syntax tree interpreter:** This kind of interpreters just need the AST to work with, so no extra compilation to an intermediate representation is needed and therefore, easier to implement. However, due it's nature, they are much slower to execute and debug due the recursiveness of working with tree structures.
- **Just-in-time compiler:** Just-in-time compilers (often called JIT compilers) are an intermediate approach between a compiler that generates machine code and an interpreter. JIT compilers compile chunks of code at a specific moments while the program run to speed up portions of the code that is beeing interpreted (for example functions that are called frequently). Essentially, the JIT compiler needs to decide when to compile a specific part of the code at runtime and adds a small overhead in exchange for a machine-language performance on specific parts.

---

<sup>2</sup>Three Address Code (Wikipedia)

## 1.4 System architecture

To design the system architecture of Nuua, consideration of existing system architectures needs to be taken since existing architectures often work better than the custom-made ones and they often lead to greater project scalability. It's trivial to make this choice before starting the project since changing a system architecture after it's initial development phases becomes a very bad choice and may lead to a big ball of mud. Two choices in software development might be hierarchical or layered systems. Nuua's architecture is based on a *layered system* [1].

A layered system has specific requirements regarding code communication. Specifically, a layered system consists of different layers arranged vertically. Those layers have a specific criterion that needs to be met. As a matter of fact, each layer can only use the layer below and gives an API for the layer above (if any) to use its functions. For example, the Figure 1.3 contains a simple 3-tier layered system. Layer 3 can only use Layer 2, and the output comes from Layer 2. Layer 3 cannot use Layer 1 nor expect any outputs from it. It's the Layer 2 responsibility to use the Layer 1 and process its output before it can give its own output.




Figure 1.3: Layered system

This system is known to be robust and scalable at a very low-performance cost. It's very easy to understand and yet very powerful for software systems like this one. These systems are also used for other complex software systems, such as operating systems or complex protocols like TCP/IP.

By using a layered system each layer gets completely isolated and works independently by just using the layer below, creating a very impressive way to scale-up or upgrade existing parts of the system without damaging the others. However, a consistent API should, in fact, be established from the ground up to avoid breaking changes. If the API

is maintained, the individual layers may be upgraded independently without the need for extra work.

The Nuua architecture is shown at Figure 1.4. An independent module called Logger is found on the left side of the figure. This module is a logger used by all layers to output messages if needed (for example error reporting).

- **Logger:** Used by all layers to debug or log errors. In case of a fatal error, the logger outputs an error stack in a fancy way and terminates the application.
- **Application:** This layer is used to decode the command line arguments and fire up the compiler toolchain. In short, the purpose is to analyze the command line arguments and fire the application accordingly.
- **Virtual Machine:** The virtual machine is the interpreter that runs Nuua. It's a register-based virtual machine that acts as the Nuua runtime environment.
- **Code generator:** Is responsible for the translation of the AST to the virtual machine bytecode. This acts as the code generation part of a compiler architecture.
- **Semantic analyzer:** Does all the semantic analysis of the compiler and optimizes the AST for faster and smaller output.
- **Parser:** Acts as the syntax analyzer, it uses a list of tokens and generates a fully valid AST.
- **Lexer:** Scans the source code and translates the characters to tokens, creating a list of tokens representing the original source code.




Figure 1.4: Nuua's architecture (Layered System)

## 2 The Nuua Language

*“Progress comes from finding better ways to do things. Don’t be afraid of innovation. Don’t be afraid of ideas that are not your own.”*

— Douglas Crockford, *December 21, 2006*

This chapter defines the Nuua programming language. The language is set to be of a general purpose usage, with an imperative paradigm and a statically typed system. The official Nuua compiler is written in C++ with a zero-dependency policy and it’s implemented as an interpreter using a register-based virtual machine.

### 2.1 Grammar

Nuua’s grammar is inspired by other existing programming languages by taking advantage of some of the best features some of them offer. Inspiration comes specially from Python, Rust and Go.

The precedence relationship between expressions is heavily inspired by C, D, Rust and Dart. The official documentation for those languages exposes similar tables for operator precedences and Nuua has taken akin levels of precedence as those.

Nuua does not make use of the ";" to separate statements, instead, it uses the same separator as Go. Statements can be separated by a "\n" but it does not make use of the "\t" to indicate statements inside blocks, and uses the typical block separator "{ ... }".

#### 2.1.1 Lexical Grammar

The lexical grammar is used by the lowest layer of the Nuua system to scan the source language and identify different terminal symbols. The main difference with the syntax grammar, as exposed in [11, Appendix I] is that the syntax grammar is a context free grammar and the lexical grammar is a regular grammar.

Nuua’s lexical rules are as follows:

DIGIT

```

: "0" ... "9"
;

```

Digits are any character from '0' to '9', given that the ASCII table <sup>1</sup> uses a sequential enumeration for the table items it's very easy to determine what characters are between '0' and '9'.

#### ALPHA

```

: "a" ... "z"
| "A" ... "Z"
| "_"
;

```

Alpha are characters that are part of the english alphabet in lower or upper case. It also includes '\_' as a special character.

#### ALPHANUM

```

: DIGIT
| ALPHA
;

```

Alphanum are characters that are either part of the alphabet or are digits.

#### INTEGER

```

: DIGIT+
;

```

Integers are a single digit or more found sequentially without spaces. The integer sign is not represented here.

#### FLOAT

```

: DIGIT+ "." DIGIT+
;

```

Floats are like integers but require a dot followed by a digit or more, creating a decimal number.

---

<sup>1</sup>ASCII table (Wikipedia)

```

BOOL
  : "true"
  | "false"
  ;

```

Bools are either 'true' or 'false', that are reserved words.

```

STRING
  : '"' @'""' '"'
  ;

```

Strings represent a character string with the possibility to escape '"' by using a '\' as a prefix, more on that in the upcoming sections.

```

IDENTIFIER
  : ALPHA ALPHANUM*
  ;

```

Identifiers are part an alpha character followed by an optional one or more alphanumeric character.

## 2.1.2 Syntax Grammar

### Program and Top Level Declarations

```

program
  : top_level_declaration*
  ;

```

A Nuua program is a list of top level declarations.

```

top_level_declaration
  : use_declaration "\n"
  | fun_declaration "\n"
  | class_declaration "\n"
  | export_declaration "\n"
  ;

```

A top level declaration can only be one of the specified rules. Top level declarations are a special type of declaration that can only be declared on the module and not inside other blocks.

```

use_declaration
  : "use" STRING
  | "use" IDENTIFIER ("," IDENTIFIER)* "from" STRING
  ;

```

A use declaration is used to import other top level declarations from other modules. By using the first rule, Nuua imports all the exported targets of the module pointed by STRING. Otherwise, Nuua imports the specified targets from the modules.

```

fun_declaration
  : "fun" IDENTIFIER "(" parameters? ")" (":" type)? fun_body;
  ;
parameters
  : variable_declaration ("," variable_declaration)*
  ;
fun_body
  : "->" expression "\n"
  | "=>" statement
  | "{" "\n" statement* "}" "\n"
  ;

```

A function is defined using the reserved word "fun" followed by the function name, and a list of optional parameters enclosed in parenthesis. The function type is specified after the parameter list by using ":". The return type might not be present if the function have no return value. The function body is specified in three different ways depending on the type of the function body that is expected.

```

class_declaration
  : "class" "{" class_statement* "}"
  ;
class_statement
  : variable_declaration
  | fun_declaration
  ;

```

A class uses a reserved word "class" and expects zero or more class statements.

```

export_declaration
  : "export" top_level_declaration
  ;

```



An export declaration marks the following top level declaration as exported, making it available for other modules to import it using the use declaration.

```
statement
  : variable_declaration "\n"?
  | if_statement "\n"
  | while_statement "\n"
  | for_statement "\n"
  | return_statement "\n"
  | print_statement "\n"
  | expression_statement "\n"?
  ;
```

Statements are used to change program's flow or indicate simple actions like declaring a variable.

```
variable_declaration
  : IDENTIFIER ":" type
  | IDENTIFIER ":" type "=" expression
  | IDENTIFIER ":" "=" expression
  ;
```

A variable declaration may have a type assigned with it, or the declaration type will be inferred from the initializer.

```
if_statement
  : "if" expression if_body el_if* else?
  ;
if_body
  : ">" statement "\n"?
  | "{" "\n" statement* "}"
  ;
el_if
  : "elif" expression if_body
  ;
else
  : "else" expression if_body
  ;
```

An "if" statement is declared with the "if" reserved word followed by the expression of its condition. The "if" body may be declared in two different ways depending on the

if body. The "elif" word may be used as a short hand to an "else" followed by an "if" inside. An optional "else" condition may be added at the end of the "if".

```
while_statement
  : "while" expression while_body
  ;
while_body
  : ">" statement "\n"?
  | "{" "\n" statement* "}"
  ;
```

A "while" statement uses the "while" reserved word followed by the expression of the condition and the "while" body, that can be specified in two different ways depending on the body contents.

```
for_statement
  : "for" IDENTIFIER ("," IDENTIFIER)? "in" expression for_body
  ;
for_body
  : ">" statement
  | "{" "\n" statement* "}"
  ;
```

A "for" statement acts as a "for in" loop in most modern high level languages. It gets declared by using the "for" reserved word followed by an identifier that will be used to store the element in the iterator. An optional second identifier may be given in case the loop index needs to be stored as well. Those values change in every iteration. After the identifiers, the "in" keyword is expected, followed by the iterator expression and the "for" body, that can be declared in two different ways depending on its content.

```
return_statement
  : "return" expression?
  ;
```

A "return" statement uses the "return" reserved word and an optional expression to return.

```
print_statement
  : "print" expression?
  ;
```

The "print" expression is a statement that outputs an expression to the screen. This is a temporary statement used while nuua is not able to properly read and write to files (stdout / stdin) in this specific case.

```
expression_statement
    : expression
    ;
```

An expression statement may be used as an expression whose value is not used or no value is returned from it.

## Expressions

Expressions are grouped in different production rules depending on their precedence, this is done due the parsing strategy used and helps to visualize the precedence by reading the grammar.

```
expression
    : assignment
    ;
```

An expression is reduced to an assignment.

```
assignment
    : range ("=" range)*
    ;
```

An assignment expression is written by an expression on the left hand side and on the right hand side with a "=" in the middle.

```
range
    : logical_or ((".." | "...") logical_or)*
    ;
```

A range expression is written by an expression on the left hand side and on the right hand side with a ".." or "..." in the middle. Depending if the range is exclusive or inclusive.

```

logical_or
  : logical_and ("or" logical_and)*
  ;

```

A logical "or" is a binary operation with the reserved word "or" in the middle.

```

logical_and
  : equality ("and" equality)*
  ;

```

A logical "and" is a binary operation with the reserved word "and" in the middle.

```

equality
  : comparison ("!=" | "==") comparison)*
  ;

```

An equality comparison is a binary operation with a "!=" or "==" in the middle depending if the check needs to be negated or not.

```

comparison
  : addition (">" | ">=" | "<" | "<=") addition)*
  ;

```

A comparison is similar to equality but checks the values to determine if the left hand side is greater, greater than, lower and lower than the right hand side.

```

addition
  : multiplication ("-" | "+") multiplication)*
  ;

```

An addition is used to perform an addition or subtraction of two values.

```

multiplication
  : cast ("/" | "*") cast)*
  ;

```

A multiplication is used to perform a multiplication or division of two values.

```

cast
  : unary_prefix ("as" type)*
  ;

```

A cast performs a type cast of the values on the left hand side to the value of the right hand side by using the "as" reserved word.

```

unary_prefix
  : ("!" | "+" | "-") unary_prefix
  | unary_postfix
  ;

```

The unary prefixes are used to change a value state by prefixing the operation.

```

unary_postfix
  : primary unary_p*
  ;
unary\_p
  : "[" expression "]"
  | slice
  | "(" arguments? ")"
  | "." IDENTIFIER;
  ;
slice
  : "[" expression? ":" expression? (":" expression?)? "]"
  ;
arguments
  : expression ("," expression)*
  ;

```

The unary postfixes are used to either access a value, to slice it's contents, to call a value or to call a value property.

```

primary
  : BOOL
  | INTEGER
  | FLOAT
  | STRING
  | IDENTIFIER
  | LIST
  | DICTIONARY

```

```

| OBJECT
| "(" expression ")"
;

```

Primary expressions have the highest precedence and are mostly native types, with the exception of the expression group.

```

OBJECT
: IDENTIFIER "{" object_args? "}"
;
object_args
: IDENTIFIER ":" expression ("," IDENTIFIER ":" expression)*
;

```

An object is defined by an identifier containing the class name, followed by optional arguments surrounded by "{" and "}" to initialize the class properties.

```

LIST
: "[" expression ("," expression)* "]"
;

```

Lists can't be empty, so a at least one expression must be provided.

```

DICTIONARY
: "{" IDENTIFIER ":" expression ("," IDENTIFIER ":" expression)* "}"
;

```

Dictionaries, as lists, can't be empty, so a at least one expression must be provided.

### 2.1.3 Operator Precedence

Given that expressions are grouped by their precedence, the operator precedence table of Nuua is as follows:

Level	Operators	Associativity
1	$A[B]$ , $A[B:C:D]$ , $A(B, C, D, \dots)$ , $A.B$	Left-to-right
2	$!A$ , $+A$ , $-A$	Right-to-left
3	$A \text{ as } B$	Left-to-right
4	$A/B$ , $A*B$	Left-to-right
5	$+A$ , $-A$	Left-to-right
6	$A > B$ , $A \geq B$ , $A < B$ , $A \leq B$	Left-to-right
7	$A \neq B$ , $A == B$	Left-to-right
8	$A \text{ and } B$	Left-to-right
9	$A \text{ or } B$	Left-to-right
10	$A..B$ , $A...B$	Left-to-right
11	$A=B$	Right-to-left

Table 2.1: Nuua operator precedence from highest to lowest with the associativity

## 2.2 Data types

This section defines all the Nuua data types that are supported. Each value in Nuua has a type associated with it, meaning that each value must belong to a certain data type. A value can't belong to multiple data types at once but can be casted to others if a change is required.

### 2.2.1 Integers

Integers are named as `int` and they are a subset of numbers represented by  $\mathbb{Z}$ . It includes 0 and the integers are stored using 64 bits. Meaning its range for a given integer  $x$  is:

$$\frac{2^{64}}{2} < x < \frac{2^{64}}{2} - 1$$

#### Examples

```
0
25
81237
-6378
-1
```

### 2.2.2 Floats

Floats are named as `float` and they are double precision points that uses a total of 64 bits. 52 fraction bits, 11 bits of exponent and 1 sign bit.

#### Examples

```
0.0
25.5
81237.11111
-6378.673
-1.9
```

### 2.2.3 Booleans

Booleans are named as `bool` and they are simple booleans, they can be either `true` or `false`.

#### Examples

```
true
false
```

### 2.2.4 Strings

Strings are named as `string` and they are used to manipulate arrays of chars. It's implementation uses a C++ `std::string`. It can store any text that's surrounded by `'"`.

#### Examples

```
"A string is represented like this"
"Another string example"
```

### 2.2.5 Lists

Lists are named as `[type]` and they are used to manipulate a list of other values. They can only have a single type as the inner list items, so all the list items need to be of the same type.



**Examples**

```
["this", "is", "a", "valid", "list", "of", "strings"]
```

**2.2.6 Dictionaries**

Dictionaries are named as `{type}` and they are used to store values of the same type. However, unlike lists, they use a string-based mapping, allowing each value to be bound to a specific string key, instead of an integer index as the key. Dictionaries, as lists, can only store 1 type of values. So each key can only store the same type.

**Examples**

```
{name: "Erik", occupation: "Student", color: "#ff0000"}
```

**2.2.7 Functions**

The only way to define a function is by using the "fun" reserved word as noted in section 2.1.2. However, functions in Nuua act as first class values, meaning that a value can contain a function, allowing the function to change without actually changing its type. The function type needs to be consistent. So even if the function changes, it will always accept the same arguments and it will return the same data type. Function types are named as `(T1, T2, ..., TN -> TR)`, where T1 to TN are the types of the function parameters. If the function has a return type, say TR, it needs to be specified with a simple arrow pointing at it the end of the function type.

To see how functions are declared head to ??

**Examples**

Function without parameters nor return type.

```
()
```

Function with 2 parameters of type int and a float return type.

```
(int, int -> float)
```

Function without parameters and a return type of a list of strings.

```
(-> [string])
```

Function with two parameters of type int and bool and no return type.

```
(int, bool)
```

### 2.2.8 Objects

Objects are named according to the class they represent. If a class is named **Person** the type name is **Person**.

## 2.3 Statements

This section explains and gives examples to all statements found in Nuua.

Some of the statements may have already been mentioned briefly in section 2.1.2.

### 2.3.1 Use Declaration

The use declaration is used when a module needs to use a top level declaration that is found in another module. The target module is the string given in the use declaration. A path system is used to search for the file, first trying to find it relatively from the current module path, ending at the standard library folder that comes with Nuua.

The use declaration comes in two different shapes. By using the use declaration with a single string it imports all the top level declarations that are exported in the target module. Instead, by determining the use identifiers, you may import only selected top level declarations.

#### Caveats

- The target module path given in the use declaration can use a relative or absolute path.
- If the target module path does not have the ".nu" extension, it will be added automatically.
- If the target module is not found, an error is thrown before any execution starts.
- If a target top level declaration is not found in the target module an error is thrown before any execution starts.
- If the top level declaration is not exported another error will be thrown prior to execution.

#### Examples

Import all exported targets in a relative file path named "test.nu"

```
use "test"
```

Import a and b from a relative file path named "test.nu"

```
use a, b from "./test.nu"
```

Import a from an absolute file path in "C:/Nuua/test.nu"

```
use a from "C:/Nuua/test.nu"
```

### 2.3.2 Function Declaration

A function declaration creates a function value and a data type given the function parameters and return value. Once the function value is created, it's then added in a new variable with the function name. That variable can be modified since functions are first class values in Nuua. The variations in a function body exists to minimize the code length in certain situations (When a function is a single return expression, a single statement or a block of statements).

#### Caveats

- Function overloading is not implemented in the latest version of Nuua as for May 29, 2019. The current plan is to have function overloading in a future version.
- Function parameters do not allow default values.
- If the function returns a value, you are expected to, at least, write a single return statement in the top level of the function block.
- Functions without return type can't be used as regular expressions since they contain no values.

#### Examples

Function without parameters and no return type.

```
fun a() {
  print "Hello, World"
}
```

Function without parameters and return type.

```
fun b(): string {
  return "Hello, World"
}
```

Function with parameters and no return type.

```
fun c(x: string) {
  print "Hello, " + x
}
```

Function with parameters and return type.

```
fun d(x: string): string {  
    return "Hello, " + x  
}
```

Single statement function.

```
fun d(x: string): string => return "Hello, " + x
```

Single expression function.

```
fun d(x: string): string -> "Hello, " + x
```

### 2.3.3 Class Declaration

A class declaration creates a data type of the given class structure. This type can then be used as a regular type to specify values of the given class. To create an object of a given class you can use the Object expression [**sec:object`expression**]. Classes act as structs with the fact that they can also contain methods bound to them. Class methods have a variable called "**self**" as a self reference to the object to mutate its state.

#### Caveats

- Class properties can't have default values. Values are defined when creating the object using an object expression.
- Self references to the same type are allowed.
- There is no class inheritance.

#### Examples

Simple class to represent a person.

```
class Person {  
    name: string  
    age: int  
    fun show() {  
        print self.name + ", " + self.age as int  
    }  
}
```

### 2.3.4 Export Declaration

The export declaration is used when a module wants to make a top level declaration available to use for other modules. Marking a top level declaration as exported allows other modules to import and use it.

#### Caveats

- You can't export another export.

#### Examples

Export a function

```
export fun add(a: int, b: int): int {  
    return a + b  
}
```

Export a class.

```
export class Person {  
    name: string  
    age: int  
}
```

### 2.3.5 Variable Declaration

Variable declarations are scoped to the lexical block where they are declared. They can be used from the block they have been declared and on the lexical blocks that may exist inside of it. A variable with the same name can't be declared in the same lexical block but multiple lexical blocks may have the same variable name. When getting the value of a variable, the lookup starts from the current block and goes backwards to previous blocks.

#### Caveats

- Even if multiple blocks have the same variable name, they all point to different values.
- If a variable is declared with an initializer, the type can be inferred by leaving the type empty.
- If a variable is declared without an initializer, the value is default initialized to a zero-state.

- If a variable is declared in a block that already contains a variable with the same name, an error is thrown before execution.

### Examples

Simple variable declaration (defaults to `int`'s zero state, in this case 0).

```
a: int
```

Variable declaration with an initializer.

```
b: int = 10
```

Variable declaration with an inferred `int` type.

```
b := 10
```

## 2.4 Expressions

Expressions...

## 3 Forthcoming Features

Although Nuua is currently in a very advanced phase, a lot of features are still missing, features that are present in most mature programming languages. This chapter defines a set of features that are planned to be implemented into the language and the compiler.

- A C foreign function interface (FFI) to call C code using Nuua. This should allow the use of libc and other existing C code.
- A proper I/O interface to read and write to files, including stdin, stdout, stderr and removing the print statement.
- Function overloading.
- Extend and create a proper usefull standard library.
- A website ([nuua.io](http://nuua.io)) to announce, showcase and write all the documentation.
- A centralized package / module manager to automate the process of using external modules at a certain version.

## 4 Bibliography

- [1] Felix Bachmann, Len Bass, Jeromy Carriere, Paul C. Clements, David Garlan, James Ivers, Robert Nord, Reed Little, *Software Architecture Documentation in Practice: Documenting Architectural Layers*, English. Software Engineering Institute, 2000. [Online]. Available: <https://resources.sei.cmu.edu/library/asset-view.cfm?assetid=5019>.
- [2] Alfred V. Aho, Monica S. Lam, Ravi Sethi, Jeffrey D. Ullman, *Compilers: Principles, Techniques, and Tools*, English, 2nd Edition. Greg Tobin, 2006, ISBN: 978-0321486813.
- [3] Terence Parr, *Language Implementation Patterns: Create Your Own Domain-Specific and General Programming Languages (Pragmatic Programmers)*, English, 1st Edition. Pragmatic Bookshelf, 2010, ISBN: 978-1934356456.
- [4] Richard Bornat, *Understanding and Writing Compilers: A do-it-yourself guide*, English, 3rd Edition. Palgrave, 1979, ISBN: 978-0333217320.
- [5] Edward Barrett, *A JIT Compiler using LLVM*, English. Bournemouth University, 2009. [Online]. Available: <http://llvm.org/pubs/2009-05-21-Thesis-Barrett-3c.pdf>.
- [6] Anders Schlichtkrull, Rasmus T. Tjalk-Bøggild, *Compiling Dynamic Languages*, English. Technical University of Denmark, 2013. [Online]. Available: [http://www2.imm.dtu.dk/pubdb/views/edoc\\_download.php/6620/pdf/imm6620.pdf](http://www2.imm.dtu.dk/pubdb/views/edoc_download.php/6620/pdf/imm6620.pdf).
- [7] Federico Tomassetti, “EBNF: How to Describe the Grammar of a Language”, English, 2017. [Online]. Available: <https://tomassetti.me/ebnf/>.
- [8] Stanford, *Three-Address Code IR*, English. [Online]. Available: <https://web.stanford.edu/class/archive/cs/cs143/cs143.1128/lectures/13/Slides13.pdf>.
- [9] Bartosz Sypytkowski, “Simple virtual machine”, English, 2014. [Online]. Available: <https://bartoszsypytkowski.com/simple-virtual-machine/>.
- [10] Andrea Bergia, “Stack Based Virtual Machines”, English, 2015. [Online]. Available: <https://andreabergia.com/stack-based-virtual-machines/>.
- [11] Bob Nystrom, *Crafting Interpreters*, English. 2018. [Online]. Available: <https://craftinginterpreters.com/>.
- [12] —, “Pratt Parsers: Expression Parsing Made Easy”, English, 2011. [Online]. Available: <https://journal.stuffwithstuff.com/2011/03/19/pratt-parsers-expression-parsing-made-easy/>.



- [13] Kein-Hong Man, esq, “A No-Frills Introduction to Lua 5.1 VM Instructions”, English, 2006. [Online]. Available: <http://luaforge.net/docman/83/98/ANoFrillsIntroToLua51VMInstructions.pdf>.
- [14] Roberto Ierusalimschy, Luiz Henrique de Figueiredo, Waldemar Celes, “The Implementation of Lua 5.0”, English, 2005. [Online]. Available: <https://www.lua.org/doc/jucs05.pdf>.
- [15] —, “Closures in Lua”, English, 2013. [Online]. Available: <https://www.cs.tufts.edu/~nr/cs257/archive/roberto-ierusalimschy/closures-draft.pdf>.